laplace (and invlaplace)

Maple has a pair of commands, `laplace` and `invlaplace` which are used to calculate Laplace transforms (and inverse Laplace transforms). The use of these commands is fairly straightforward -- Maple knows the formulas in the standard tables of Laplace transforms, as well as the operational properties (convolutions, derivatives, products, etc..), so these two commands can be applied to equations as well as to expressions.

The use of the commands is easiest to understand based on examples. But before we use them, because they are in Maples integral transforms library (called `inttrans`), we must execute the command:

```
> with(inttrans):
Warning, new definition for hilbert
```

Then, to calculate the Laplace transform of the expression `t^3`, we enter:

```
> laplace(t^3,t,s);
```

$$\frac{6}{s^4}$$

The syntax of the command requires three things between the parentheses: first, the expression whose Laplace transform is being taken, second, the name of the variable in the expression (in case other parameters are involved) -- usually this variable is t or x. Last is the name of the variable to be used in the expression of the Laplace transform (usually s or p) -- any variable may be used, as long as it doesn't conflict with a name already in use.

It is possible to take the Laplace transform of differential or integral equations -- when doing so it is important to note that, just as with `dsolve`, it is important to indicate which letters stand for functions by writing, for instance `y(t)` rather than just `y`:

```
> laplace(diff(y(t),t)=3*y(t)+exp(-t),t,s);
```

$$s\ \mathrm{laplace}(\mathrm{y}(t), t, s) - \mathrm{y}(0) = 3\ \mathrm{laplace}(\mathrm{y}(t), t, s) + \frac{1}{s+1}$$

Notice that in this case, Maple has applied two (or three) of the operational properties of Laplace transforms -- that of derivatives on the left side, and the one for exponentials and shift operators on the right. One way to proceed with solving the differential equation is to solve the algebraic equation above for `laplace(y(t),t,s)` :

```
> solve(",laplace(y(t),t,s));
```

$$\frac{\mathrm{y}(0)\,s + \mathrm{y}(0) + 1}{s^2 - 2\,s - 3}$$

and then to take the inverse Laplace transform with invlaplace:

```
> invlaplace(",s,t);
```

$$-\frac{1}{4}\mathbf{e}^{(-t)} + \mathbf{e}^{(3\,t)}\,\mathrm{y}(0) + \frac{1}{4}\mathbf{e}^{(3\,t)}$$

This gives the solution in terms of the initial condition. On the other hand, the simplest way to get Maple to solve the differential equation in preceding example is to use the `dsolve` command with the "method=laplace" option, as follows:

```
> dsolve(diff(y(t),t)=3*y(t)+exp(-t),y(t),method=laplace);
```

$$\mathrm{y}(t) = -\frac{1}{4}\mathbf{e}^{(-t)} + \mathbf{e}^{(3\,t)}\,\mathrm{y}(0) + \frac{1}{4}\mathbf{e}^{(3\,t)}$$

As you undoubtedly already noticed, the `invlaplace` command undoes Laplace transforms -- we must tell this command about the variables, too but in the opposite order:

```
> invlaplace(2/(s^2+5),s,t);
```

$$\frac{2}{5}\sqrt{5}\,\sin(\sqrt{5}\,t)$$

Two functions that come up in the context of working with Laplace transforms are the Dirac delta function and the unit step function (or Heaviside function). Maple understands these functions and their Laplace transforms. Their names in Maple are `Dirac(t)` and `Heaviside(t)` respectively:

```
> plot(Heaviside(t-3),t=0..5);
```




```
> laplace(Dirac(t-4),t,s);
```

$$e^{(-4\,s)}$$

```
> diff(Heaviside(t-2),t);
```

$$\mathrm{Dirac}(t-2)$$